libname mylib 'D:\Dropbox\Wall Street Bets (Private)\Data';

*This program contains the information needed to create Table 7 of the paper;
*The program references daily_panel_rfs whicht is constructed in  Daily Panel RFS (Intermediate File) code;


*The code below reports specification 1 of Table 5 (i.e., 5-day ahead media sentiment).


* To obtain Specification 2, replace Sentiment 5 with sentiment 21;

*To obstain Specification 3, replace Sentiment 5 with beat and impose the filter that there is at least one forecast ("ln_forecasts >0") 
and limit to forecasts  issued within one week of earnings ("what 1<=date<=dif<=7);

*To obstain Specification 4, replace Sentiment 5 with beat and impose the filter that there is at least one forecast ("ln_forecasts >0") 
and limit to forecasts issued within four week of earnings ("what 1<=date<=dif<=28);

*To obstain Specification 5, replace Sentiment 5 with percent_up5b and impose the filter that there is at least one forecast ("ln_forecasts >0") 

*To obstain Specification 6, replace Sentiment 5 with percent_up21b and impose the filter that there is at least one forecast ("ln_forecasts >0");





data info2;
set mylib.daily_panel_rfs;;

keep ticker date cum_month gme_amc_flag post_gme
ret5 ret21
net_dd2  NET_DD2_POST NET_SA2 NET_SA2_POST NON_RESEARCH2 NON_RESEARCH2_POST
DD_and_SA DD_and_SA_POST DD_and_Non_Research DD_and_Non_Research_post net_dd2_pre
 ln_size ln_bm  abn_ret mom5 mom6_26     
BM_MISSING SIZE_MISSING   news_sentiment lag_sent5 lag_sent6_26 
sentiment5 SENTIMENT21
percent_up5b percent_up21b 
beat date_dif LN_FORECASTS;
run;


PROC SORT DATA=info2;
BY DATE;
RUN;
PROC STANDARD DATA=info2 OUT=INFO_FE MEAN=0;
BY DATE;
var ret5 ret21
net_dd2  NET_DD2_POST NET_SA2 NET_SA2_POST NON_RESEARCH2 NON_RESEARCH2_POST
DD_and_SA DD_and_SA_POST DD_and_Non_Research DD_and_Non_Research_post net_dd2_pre
 ln_size ln_bm  abn_ret mom5 mom6_26     
BM_MISSING SIZE_MISSING   news_sentiment lag_sent5 lag_sent6_26 sentiment5 SENTIMENT21
 percent_up5b percent_up21b
beat ;
*WHERE LN_FORECASTS >0 ;
*where 1<=date_dif<=28 and ln_forecasts >0;
RUN;





%let yourdata=INFO_FE;

%let firmid=ticker ;
%let time=cum_month;

%let y=sentiment5; ;
%let x=  net_dd2  NET_DD2_POST  NON_RESEARCH2 NON_RESEARCH2_POST NET_SA2 NET_SA2_POST ln_size ln_bm  abn_ret mom5 mom6_26       news_sentiment lag_sent5 lag_sent6_26   BM_MISSING SIZE_MISSING;
RUN;




proc surveyreg data=&yourdata;

    cluster &firmid;
    model &y = &x /covb;
    ods output covb=firm;
run;
quit;


*cluster by second dimension (e.g., year);
proc surveyreg data=&yourdata;

    cluster &time;
    model &y = &x /covb;
    ods output covb=year;
run;
quit;

*cluster by intersection of the two dimensions (e.g, firm-year);
proc surveyreg data=&yourdata;
    cluster &firmid &time;
    model &y = &x /covb;
    ods output covb=both;
    ods output parameterestimates=parm;
run;
quit;

*keeps original parameter estimates;
data parm; set parm;
    keep parameter estimate;
run;

*returns a dataset with a scalar for the dimensions of the var/cov matrix. This is needed to extract the square roots of the diagonals later on;
data parm1; set parm;
    n=_n_;
    m=1;
    keep m n;
run;

data parm1; set parm1; by m;
    if last.m;
    keep n;
run;

*uses matrix algebra interface to construct Var-cov matrix and extract the standard errors;
proc iml;
    use both;
    read all var _num_ into Z;
    print Z;
    use firm;
    read all var _num_ into X;
    print X;
    use year;
    read all var _num_ into Y;
    print Y;
    use parm1;
    read all var _num_ into n;
    print n;

    B=X+Y-Z;
    C=I(n);
    D=J(n,1);
    E=C#B;
    F=E*D;
    G=F##.5;

    print B;
    print G;
    create b from G [colname='stderr']; ;
    append from G;
quit;

*creates a dataset called 'results' that contains the parameter estimates, the SE's, and the t-stats;
data results; merge parm B;
    tstat=estimate/stderr;
run;

proc print data=results;
run;




data spec3;
set results;
estimate3 = estimate  ;
obs = _n_;
keep parameter estimate3 obs;
run;
data spec3b;
set results;
estimate3 = tstat ;
obs = _n_ + .5;
keep parameter estimate3 obs;
run;
data spec3;
set spec3 spec3b;
run;
proc sort data=spec3;
by obs;
run;
